

Skalarprodukt, Hessesche Normalform

Prof. Dr. Dörte Haftendorn: Mathematik mit MuPAD 4.0, (MuPAD 3 im Okt. 05) Update 8.06.07

<http://haftendorn.uni-lueneburg.de>

www.mathematik-verstehen.de

```
o:=matrix([0,0,0]):  
a:=matrix([4,2,-1]):  
v:=matrix([3,1,-4]):  
w:=matrix([2,3,-5]): a,v,w
```


$$\begin{pmatrix} 4 \\ 2 \\ -1 \end{pmatrix}, \begin{pmatrix} 3 \\ 1 \\ -4 \end{pmatrix}, \begin{pmatrix} 2 \\ 3 \\ -5 \end{pmatrix}$$

Ebene

```
ebene:=a+ r*v+s*w
```

$$\begin{pmatrix} 3 \cdot r + 2 \cdot s + 4 \\ r + 3 \cdot s + 2 \\ -r \cdot 4 - s \cdot 5 - 1 \end{pmatrix}$$

```
apkt:=plot::Point3d(a):  
av:=plot::Arrow3d(o,a):  
vv:=plot::Arrow3d(a,a+v, LineColor=[0,1,0]):  
ww:=plot::Arrow3d(a,a+w):  
eb:=plot::Surface(ebene,r=-5..5,s=-5..5,  
ViewingBox=[-5..8,-5..6,-8..5]):  
plot(eb,apkt,ww,vv,av, Scaling=Constrained)
```



Skalarprodukt

```
vw:=linalg::scalarProduct(v,w);  
lvq:=linalg::scalarProduct(v,v);  
lwq:=linalg::scalarProduct(w,w);
```

1

29

26

38

Berechnung der Winkels zwischen v und w

```
cosi:=vw/sqrt(lvq*lwq); float(cosi);  
float(arccos(cosi)*180/PI)
```

$$\frac{29 \cdot \sqrt{247}}{494}$$

0.9226129063

22.68888931


Das ist der Winkel im Gradmaß

Kreuzprodukt

```
n:=linalg::crossProduct(v,w);  
nv:=plot::Arrow3d(a,n, LineColor=[1,0,0]):
```

$$\begin{pmatrix} 7 \\ 7 \\ 7 \end{pmatrix}$$

```
plot(eb,nv,av,vv,ww,Scaling=Constrained)
```



Hessesche Normalenform

```
p:=matrix([x,y,z]):  
ehess:=(linalg::scalarProduct((p-a),n)=0)
```

$$7 \cdot x + 7 \cdot y + 7 \cdot z - 35 = 0$$

$$7 \cdot x + 7 \cdot y + 7 \cdot z - 35 = 0$$

```
eh:=plot::Implicit3d(ehess,x=-5..8,y=-5..5,z=-8..5):  
plot(eh,av,nv,vv,ww, Scaling=Constrained)
```

